
Integracja z WebAPI 
Elektronicznego Nadawcy 
z poziomu języka Visual Basic 
i oprogramowania Visual Studio 2019


1. Framework .NET Framework 4.8 i niższe
1.1. Tworzenie klasy z referencjami
W tych frameworkach można skorzystać z kreatora do wygenerowania klasy z referencjami do webserwisu:
· [bookmark: _GoBack]klikamy prawym przyciskiem myszy na projekcie w Visual Studio i wybieramy Add → Service Reference…
· W oknie Add Service Reference klikamy w przycisk Advanced… w lewym dolnym rogu

· W oknie Service Reference Settings klikamy przycisk Add Web Reference… w lewym dolnym rogu
[image: ]
· W oknie Add Web Reference w polu URL podajemy adres sieciowy (lub lokalizacja pliku na dysku) z którego można pobrać plik WSDL dla webserwisu (adres do pobrania dla webserwisu testowego to https://en-testwebapi.poczta-polska.pl/websrv/labs.php?wsdl zaś dla webserwisu produkcyjnego to https://e-nadawca.poczta-polska.pl/websrv/labs.php?wsdl) i naciskamy przycisk strzałki w celu załadowania pliku WSDL. Następnie w polu Web  Reference Name wpisujemy namespace (np. labs) dla pliku z referencjami i naciskamy przycisk Add Reference.

· W tym momencie w projekcie z naszą aplikacją powinna pojawić się referencja labs

1.2. Przykładowy kod źródłowy
Inicjalizacja klienta webserwisu
' włączenie TLS 1.2
ServicePointManager.SecurityProtocol = SecurityProtocolType.Tls12

Dim tEN As labs.ElektronicznyNadawca = New labs.ElektronicznyNadawca
tEN.Url = "https://en-testwebapi.poczta-polska.pl/websrv/labs.php"
'tEN.Url = "https://e-nadawca.poczta-polska.pl/websrv/labs.php"
Dim c As NetworkCredential = New NetworkCredential
c.UserName = "username"
c.Password = "password"
Dim cc As CredentialCache = New CredentialCache
cc.Add(New Uri("https://en-testwebapi.poczta-polska.pl"), "Basic", c)
'cc.Add(New Uri("https://e-nadawca.poczta-polska.pl"), "Basic", c)
tEN.Credentials = cc

Gdzie username i password są to nasze dane otrzymane po zarejestrowaniu się na środowisku testowym https://en-testwebapi.poczta-polska.pl lub produkcyjnym https://e-nadawca.poczta-polska.pl. W zależności od używanego środowiska musimy też podać adres endpointa testowego https://en-testwebapi.poczta-polska.pl/websrv/labs.php lub produkcyjnego https://e-nadawca.poczta-polska.pl/websrv/labs.php. W obiekcie klasy CredentialCache podajemy prefix adresu endpointa.
Wykonanie metody testowej hello
Dim test As String = tEN.hello("test")

Przekazanie danych o przesyłce (na przykładzie paczki pocztowej)
Dim przes As List(Of labs.przesylkaType) = New List(Of labs.przesylkaType)
Dim paczka As labs.paczkaPocztowaType = New labs.paczkaPocztowaType
paczka.masa = 20000
paczka.guid = Guid.NewGuid().ToString()
paczka.adres = New labs.adresType
paczka.adres.nazwa = "Jan Kowalski"
paczka.adres.ulica = "Woronicza"
paczka.adres.numerDomu = "12"
paczka.adres.miejscowosc = "Warszawa"
paczka.adres.kodPocztowy = "00950" 'ważne 
paczka.adres.telefon = "501123456"
paczka.kategoria = labs.kategoriaType.PRIORYTETOWA
paczka.gabaryt = labs.gabarytType.GABARYT_A
przes.Add(paczka)
'{…inne parametry…} 
Dim resp As labs.addShipmentResponseItemType() = tEN.addShipment(przes.ToArray(), 0, False)
Dim test2 As String = resp(0).numerNadania

Pobranie danych o możliwych urzędach nadania
Dim uNadania As labs.urzadNadaniaFullType() = tEN.getUrzedyNadania()

Wysłanie danych o przesyłkach do urzędu
Dim pakiet As labs.pakietType() = Nothing
Dim profil As labs.profilType = Nothing
Dim idEnvelope As Integer
Dim idEnvelopeSpec As Boolean
Dim envelopeStatus As labs.envelopeStatusType
Dim envelopeStatusSpec As Boolean
Dim bladWywolania As labs.errorType() = Nothing
Dim envelopeFilename As String = Nothing
tEN.sendEnvelope(uNadania(0).urzadNadania, True, pakiet, 0, False, profil, idEnvelope, idEnvelopeSpec, envelopeStatus, envelopeStatusSpec, bladWywolania, envelopeFilename)
Dim test3 As String = envelopeStatus.ToString()

Pobranie informacji o statusie pakietu
Dim envelopeStatus2 As labs.envelopeStatusType
Dim envelopeStatusSpec2 As Boolean
Dim bladWywolania2 As labs.errorType() = Nothing
tEN.getEnvelopeStatus(idEnvelope, envelopeStatus2, envelopeStatusSpec2, bladWywolania2)


2. Framework .NET Core 3.1 i .NET 5.0
2.1. Tworzenie klasy z referencjami
W tych frameworkach trzeba ręcznie wygenerować klasę z referencjami do webserwisu za pomocą narzędzia svcutil.exe. 
Należy sprawdzić czy mamy go zainstalowanego wraz z pakietem Windows SDK. Przykładowa ścieżka
C:\Program Files (x86)\Microsoft SDKs\Windows\v10.0A\bin\NETFX 4.8 Tools\SvcUtil.exe
Jeżeli już mamy go zainstalowanego to w konsoli, w ścieżce gdzie znajduje się svcutil.exe wydajemy polecenie
svcutil.exe /wrapped /serializer:Auto /language:vb /out:"[ścieżka_katalogu_projektu]\Service References\labs\Reference.vb" /config:"[ścieżka_katalogu_projektu]\Service References\labs\output.config" /namespace:*,labs [adres_pliku_wsdl_webserwisu]
gdzie:
[ścieżka_katalogu_projektu] – nazwa katalogu w którym znajduje się projekt naszej aplikacji w Visual Studio,
[adres_pliku_wsdl_webserwisu] – adres sieciowy (lub lokalizacja pliku na dysku) z którego można pobrać plik WSDL dla webserwisu, adres do pobrania dla webserwisu testowego to
https://en-testwebapi.poczta-polska.pl/websrv/labs.php?wsdl
zaś dla webserwisu produkcyjnego to
https://e-nadawca.poczta-polska.pl/websrv/labs.php?wsdl
Przykład dla webserwisu testowego
svcutil.exe /wrapped /serializer:Auto /language:vb /out:"C:\Projekty\integracjaen\n50\Service References\labs\Reference.vb" /config:"C:\Projekty\integracjaen\n50\Service References\labs\output.config" /namespace:*,labs https://en-testwebapi.poczta-polska.pl/websrv/labs.php?wsdl
Po wydaniu polecenia w katalogu z naszą aplikacją powinien pojawić się katalog zawierający plik Reference.vb, po dodaniu tego katalogu do projektu powinno to wyglądać następująco:

2.2. Dodawanie wymaganych pakietów NuGet
Aby można było skompilować plik Reference.vb do projektu z naszą aplikacją należ dodać następujące pakiety NuGet w wersji zgodnej z używanym frameworkiem:
· System.ServiceModel.Primitives
· System.ServiceModel.Http

2.3. Przykładowy kod źródłowy
Inicjalizacja klienta webserwisu
' włączenie TLS 1.2
ServicePointManager.SecurityProtocol = SecurityProtocolType.Tls12

Dim tEN As labs.ElektronicznyNadawcaClient = New labs.ElektronicznyNadawcaClient(
    GetBinding(),
    New EndpointAddress("https://en-testwebapi.poczta-polska.pl/websrv/labs.php"))
'Dim tEN As labs.ElektronicznyNadawcaClient = New labs.ElektronicznyNadawcaClient(
'   GetBinding(),
'   New EndpointAddress("https://e-nadawca.poczta-polska.pl/websrv/labs.php"))
tEN.ClientCredentials.UserName.UserName = "username"
tEN.ClientCredentials.UserName.Password = "password"

Gdzie username i password są to nasze dane otrzymane po zarejestrowaniu się na środowisku testowym https://en-testwebapi.poczta-polska.pl lub produkcyjnym https://e-nadawca.poczta-polska.pl. W zależności od używanego środowiska musimy też podać adres endpointa testowego https://en-testwebapi.poczta-polska.pl/websrv/labs.php lub produkcyjnego https://e-nadawca.poczta-polska.pl/websrv/labs.php.
Metoda GetBinding ustawia domyśle wartości powiązania
Function GetBinding() As Binding
    Dim result As BasicHttpBinding = New BasicHttpBinding
    result.MaxBufferSize = Integer.MaxValue
    result.ReaderQuotas = System.Xml.XmlDictionaryReaderQuotas.Max
    result.MaxReceivedMessageSize = Integer.MaxValue
    result.AllowCookies = True
    result.Security.Mode = BasicHttpSecurityMode.Transport

    Return result
End Function

Wykonanie metody testowej hello
Using scope As New OperationContextScope(tEN.InnerChannel)
    SetAuthorizationHeader(tEN.ClientCredentials.UserName)

    ' Wykonanie metody testowej
    Dim test As String = tEN.hello(New labs.hello With {.[in] = "test"}).out
End Using

Do każdego żądania wysyłanego do webserwisu musimy dołączyć nagłówek z autoryzacją za pomocą metody SetAuthorizationHeader
Sub SetAuthorizationHeader(username As UserNamePasswordClientCredential)
    Dim httpRequestProperty As HttpRequestMessageProperty = New HttpRequestMessageProperty
    Dim encodedCredentials As String = Convert.ToBase64String(
        Encoding.ASCII.GetBytes(username.UserName + ":" + username.Password))
    httpRequestProperty.Headers(HttpRequestHeader.Authorization) = "Basic " + encodedCredentials
    OperationContext.Current.OutgoingMessageProperties(HttpRequestMessageProperty.Name) = httpRequestProperty
End Sub

Przekazanie danych o przesyłce (na przykładzie paczki pocztowej)
Using scope As New OperationContextScope(tEN.InnerChannel)
    SetAuthorizationHeader(tEN.ClientCredentials.UserName)

    ' Przekazanie danych o przesyłce (na przykładzie paczki pocztowej)
    Dim przes As List(Of labs.przesylkaType) = New List(Of labs.przesylkaType)
    Dim paczka As labs.paczkaPocztowaType = New labs.paczkaPocztowaType
    paczka.masa = 20000
    paczka.guid = Guid.NewGuid().ToString()
    paczka.adres = New labs.adresType
    paczka.adres.nazwa = "Jan Kowalski"
    paczka.adres.ulica = "Woronicza"
    paczka.adres.numerDomu = "12"
    paczka.adres.miejscowosc = "Warszawa"
    paczka.adres.kodPocztowy = "00950" 'ważne 
    paczka.adres.telefon = "501123456"
    paczka.kategoria = labs.kategoriaType.PRIORYTETOWA
    paczka.gabaryt = labs.gabarytType.GABARYT_A
    przes.Add(paczka)
    '{…inne parametry…} 
    Dim addShipment1 As labs.addShipment = New labs.addShipment With
    {
        .przesylki = przes.ToArray(),
        .idBufor = 0,
        .idBuforSpecified = False
    }
    Dim resp As labs.addShipmentResponseItemType() = tEN.addShipment(addShipment1)
    Dim test2 As String = resp(0).numerNadania
End Using

Pobranie danych o możliwych urzędach nadania
Dim uNadania As labs.urzadNadaniaFullType()
Using scope As New OperationContextScope(tEN.InnerChannel)
    SetAuthorizationHeader(tEN.ClientCredentials.UserName)

    ' Pobranie danych o możliwych urzędach nadania
    uNadania = tEN.getUrzedyNadania(New labs.getUrzedyNadania())
End Using

Wysłanie danych o przesyłkach do urzędu
Dim idEnvelope As Integer
Using scope As New OperationContextScope(tEN.InnerChannel)
    SetAuthorizationHeader(tEN.ClientCredentials.UserName)

    ' Wysłanie danych o przesyłkach do urzędu 
    Dim sendEnvelope1 As labs.sendEnvelope = New labs.sendEnvelope With
    {
        .idBufor = 0,
        .idBuforSpecified = False,
        .pakiet = Nothing,
        .profil = Nothing,
        .urzadNadania = uNadania(0).urzadNadania,
        .urzadNadaniaSpecified = True
    }
    Dim resp As labs.sendEnvelopeResponseType = tEN.sendEnvelope(sendEnvelope1)
    Dim test3 As String = resp.envelopeStatus.ToString()
    idEnvelope = resp.idEnvelope
End Using

Pobranie informacji o statusie pakietu
Using scope As New OperationContextScope(tEN.InnerChannel)
    SetAuthorizationHeader(tEN.ClientCredentials.UserName)

    ' Pobranie informacji o statusie pakietu
    Dim getEnvelopeStatus1 As labs.getEnvelopeStatus = New labs.getEnvelopeStatus With
    {
        .idEnvelope = idEnvelope
    }
    Dim resp As labs.getEnvelopeStatusResponse = tEN.getEnvelopeStatus(getEnvelopeStatus1)
    Dim test4 As String = resp.envelopeStatus.ToString()
End Using

Navigate to a web service URL and click Add Reference to add allthe available services.

]} Web services found at this URL:

<?xml version="1.0" encoding="UTF-8"?>
<wsdl:definitions
targetiamespace="http:/ /e-nadawca.poczta-polska.pl”
nam
xmins: http://www.w3.0rg/2001/XMLSchema"
xmins: "http://schemas.xmlsoap.org/wsdl/"
xmins: 'http://e-nadawca.poczta-polska.pl
xmins: "http://schemas.xmlsoap.org/wsdl/soap/">
- <wsdi:types>
- <xsd:schema
targetamespace="http://e-nadawca.poczta-
polska.pl'>
- <xsd:element name="addShipment">
- <xsd:complexType>
- <xsd:sequence>
<xsd:element name="przesylki
maxOccurs="500" minoccurs:
type="tns:przesylkaType"/>
<xsd:element name="idBufor"
maxOccurs="1" minOccur:
type="xsd:int"/>
</xsd:sequence>

